PRACTICA 3b

Resolución de sistemas lineales. Métodos iterativos.

Métodos iterativos. Un método iterativo para resolver el sistema lineal $A\mathbf{x} = \mathbf{b}$ comienza con una aproximación inicial $\mathbf{x}^{(0)}$ a la solución $\mathbf{x}$ y genera una sucesión de vectores $\{\mathbf{x}^{(k)}\}_{k=0}^{\infty}$ que se espera converja a $\mathbf{x}$. Los dos métodos iterativos básicos son el *método de Jacobi* y el *método de Gauss–Seidel* cuyas implementaciones algorítmicas se describen a continuación.

Método de Jacobi

Dada A de $n \times n$ con elementos diagonales no nulos.
Escoger $\mathbf{x}^{(0)}$ de orden n, usualmente $\mathbf{x}^{(0)} = \mathbf{0}$.
Para $k = 1, 2, \ldots$

$$\text{Tomar } x_i^{(k)} = \frac{b_i - \sum_{\substack{j=1 \\ (j \neq i)}}^{n} a_{ij} x_j^{(k-1)}}{a_{ii}}, \quad \text{para } i = 1, \ldots, n.$$

Método de Gauss–Seidel

Dada A de $n \times n$ con elementos diagonales no nulos.
Escoger $\mathbf{x}^{(0)}$ de orden n, usualmente $\mathbf{x}^{(0)} = \mathbf{0}$.
Para $k = 1, 2, \ldots$

$$\text{Tomar } x_i^{(k)} = \frac{b_i - \sum_{j=1}^{i-1} a_{ij} x_j^{(k)} - \sum_{j=i+1}^{n} a_{ij} x_j^{(k-1)}}{a_{ii}}, \quad \text{para } i = 1, \ldots, n.$$

Como en todo método iterativo, debe establecerse un criterio de paro para las iteraciones, el cual, en este contexto, puede ser

$$\frac{\|\mathbf{x}^{(k)} - \mathbf{x}^{(k-1)}\|}{\|\mathbf{x}^{(k)}\|} < \epsilon,$$

respecto de una norma (por ejemplo l_∞) y cierta tolerancia ϵ prefijada.

☞ Si A es invertible, la condición $a_{ii} \neq 0$ siempre puede ser lograda con un apropiado intercambio de filas.

☞ Nótese que en el método de Jacobi cada componente de una nueva iteración es generada a partir de las componentes de la iteración anterior en forma independiente y simultáneamente, de aquí que el método se conozca también como *método de desplazamientos simultáneos*. En el método de Gauss-Seidel, por otra parte, las componentes de una nueva iteración son sucesivamente actualizadas tan pronto una es conocida, y por ello al método se lo conoce también como *método de desplazamientos sucesivos*.

☞ Desde el punto de vista de la implementación computacional, lo anterior implica que el método de Jacobi requiere del almacenamiento de dos vectores, el de la iteración actual y el de la anterior, en tanto que el método de Gauss-Seidel permite considerar un solo vector cuyas componentes se actualizan al momento de la iteración. Por otra parte, la naturaleza simultánea de las correcciones en el método de Jacobi, hacen del mismo un método inherentemente *paralelizable*, mientras que, en principio, el método de Gauss-Seidel es *secuencial*[1].

☞ En el módulo `iterative`, presentado en el código 1, el método de Jacobi y de Gauss-Seidel son implementados por las subrutinas `jacobi` y `gauss_seidel`, respectivamente.

Ejercicio 1. El sistema

$$\begin{aligned} 2\,x_1 - x_2 &= 1 \\ -x_1 + 2\,x_2 &= 1 \end{aligned}$$

tiene la solución $\mathbf{x} = (1, 1)^{\mathrm{T}}$. Considere tres iteraciones *a mano* del método de Jacobi y del método de Gauss-Seidel.

Ejercicio 2. Escribir programas Fortran para resolver un sistema de ecuaciones lineales con las implementaciones del método de Jacobi y el método de Gauss-Seidel del módulo `iterative` dadas en el código 1. Testee los mismos con el sistema del ejercicio anterior.

Ejercicio 3. Utilizar los programas anteriores para resolver los siguientes sistemas de ecuaciones:

$$a)\quad \begin{aligned} 3.8\,x_1 &+ 1.6\,x_2 &+ 0.9\,x_3 &= 15.5 \\ -0.7\,x_1 &+ 5.4\,x_2 &+ 1.6\,x_3 &= 10.3 \\ 1.5\,x_1 &+ 1.1\,x_2 &- 3.2\,x_3 &= 3.5 \end{aligned}$$

$$b)\quad \begin{aligned} x_1 &+ x_3 &= 4 \\ -x_1 &+ x_2 &= 1 \\ x_1 &+ 2x_2 &- 3x_3 &= -4 \end{aligned}$$

$$c)\quad \begin{aligned} x_1 &+ 0.5\,x_2 &+ 0.5\,x_3 &= 2 \\ 0.5\,x_1 &+ x_2 &+ 0.5\,x_3 &= 2 \\ 0.5\,x_1 &+ 0.5\,x_2 &+ x_3 &= 2 \end{aligned}$$

Convergencia. Los sistemas del ejercicio anterior muestran que los métodos iterativos no siempre son convergentes. La condición *necesaria y suficiente* para la convergencia es dada como sigue. Sea A la matriz cuadrada de orden n del sistema. Denotemos por D a la matriz diagonal de orden n formada con los elementos de la diagonal principal de A y ceros en los demás elementos. Sea $-L$ la matriz estrictamente triangular inferior de orden n formada con los elementos de A situados bajo la diagonal principal y ceros en los demás elementos. Sea $-U$ la matriz estrictamente triangular superior de orden n formada con los elementos situados por arriba de la diagonal principal y ceros en los demás elementos. Entonces

$$A = D - L - U.$$

Sea

$$T = \begin{cases} D^{-1}(L+U) & \text{para el método de Jacobi,} \\ (D-L)^{-1}U & \text{para el método de Gauss–Seidel.} \end{cases}$$

[1] Sin embargo, estableciendo ciertos reordenamientos de las ecuaciones puede lograrse una paralelización del método de Gauss-Seidel.

Entonces el método iterativo respectivo converge para cualquier elección del vector inicial, si y solo si, $\rho(T) < 1$. Aquí $\rho(T)$ denota el *radio espectral* de T, esto es, $\rho(T) = \max\{|\lambda_i|, \lambda_i \text{ autovalor de } T\}$. Además, la velocidad de convergencia del método depende de $\rho(T)$, cuanto menor sea este valor más rápida será la convergencia.

Condiciones suficientes, pero *no* necesarias para la convergencia de los métodos son las siguientes:

☞ Si para alguna norma natural $\|T\| < 1$ entonces el método iterativo correspondiente converge.

☞ Si A es una matriz *estrictamente diagonal dominante (por filas)*, esto es,

$$|a_{ii}| > \sum_{\substack{j=1 \\ j\neq i}}^{n} |a_{ij}|, \qquad \text{para } i = 1, \ldots, n$$

entonces tanto el método de Jacobi como el método de Gauss–Seidel convergen.

☞ Si A es simétrica y definida positiva el método de Gauss-Seidel converge.

Ejercicio 4. Justificar la convergencia (o divergencia) de los métodos iterativos para los sistemas del ejercicio anterior.

Matrices ralas. Una matriz A de $n \times n$ se dice que es *rala* (o *dispersa*), si sólo una pequeña fracción de sus elementos son no nulos. El correspondiente sistema lineal $A\mathbf{x} = \mathbf{b}$ se dice que es ralo si la matriz A es rala (nótese que el término independiente $\mathbf{b}$ puede ser denso o ralo). Un simple caso de matrices ralas lo constituyen las matrices de banda con anchos de banda $w \ll n$. Pero en un caso más general, los elementos no nulos pueden estar distribuidos de una manera menos sistemática. Cuando se resuelve un sistema lineal ralo sólo los elementos no nulos de A deben ser almacenados y el método utilizado debe evitar operar con elementos que se saben que serán nulos. La implementación de un método directo para resolver un sistema ralo en forma eficiente conduce a algoritmos complejos. Para los métodos iterativos, en cambio, puesto que éstos trabajan directamente con la matriz A e involucran términos del tipo producto de una matriz por un vector, la implementación es computacionalmente más simple y el costo computacional en cada iteración es proporcional al número de elementos no nulos de A. En lo que sigue consideraremos la adaptación del método de Jacobi a un sistema ralo.

En primer lugar hay que considerar un esquema de almacenamiento de la matriz rala A que permita almacenar sólo los elementos no nulos. Aquí consideraremos el esquema más simple, denominado *representación de coordenadas*. Sea nz el número de elementos *no nulos* de A, entonces A es representada por tres arreglos unidimensionales de nz elementos:



Figura 1. Distribución de los elementos no nulos de la matriz del ejercicio 5.

un arreglo real **aa** conteniendo los elementos reales no nulos de A *en cualquier orden* y dos arreglos enteros **ia** y **ja** conteniendo los índices de la fila y columna, respectivamente, de cada elemento almacenado en **aa**. Por ejemplo, la matriz de 5×5:

$$\begin{pmatrix} 11 & 12 & 0 & 0 & 15 \\ 21 & 22 & 0 & 0 & 0 \\ 0 & 0 & 33 & 0 & 35 \\ 0 & 0 & 0 & 44 & 0 \\ 51 & 0 & 53 & 0 & 55 \end{pmatrix}$$

con $nz = 11$ elementos no nulos, puede ser almacenada como

```
aa= [51,12,11,33,15,53,55,22,35,44,21]
ia= [ 5, 1, 1, 3, 1, 5, 5, 2, 3, 4, 2]
ja= [ 1, 2, 1, 3, 5, 3, 5, 2, 5, 4, 1]
```

Por otra parte, el método de Jacobi requiere, en cada iteración, una operación del tipo $\mathbf{y} = A\mathbf{x}$. Es fácil ver que, en la representación de coordenadas de A, los elementos de este producto pueden calcularse según:

```
y = 0.0_wp
do k = 1,nz
  y(ia(k)) = y(ia(k)) + aa(k)*x(ja(k))
end do
```

La implementación del método de Jacobi para matrices ralas, en la representación de coordenadas, está dada por la subrutina `jacobi_sparse` dentro del módulo `iterative` del código 1.

Ejercicio 5. Resuelva el sistema $A\mathbf{x} = \mathbf{b}$ de orden $n = 312$ cuya matriz A tiene $nz = 1874$ elementos no nulos almacenados, en la representación de coordenadas, en el archivo sistema.dat, estando $\mathbf{b}$ almacenado a continuación en el mismo archivo. *Nota:* la solución exacta del sistema tiene la forma $\mathbf{x} = (0, 1, 2, \ldots, 8, 9, 0, 1, 2, \ldots, 8, 9, 0, 1, 2, \ldots)^{\mathrm{T}}$. La fig. 1 muestra la distribución de los elementos no nulos de la matriz del sistema (el gráfico fue obtenido con la función `spy` del programa `octave`).

Código 1. Implementación de los métodos iterativos para la resolución de sistemas de ecuaciones lineales

```
module iterative

  use iso_fortran_env, only: wp => real64
  implicit none

  private :: matvec_sparse

contains

  subroutine jacobi(a,b,x,tol,iter,clave)

    ! METODO DE JACOBI para la resolución por iteración del
    ! sistema de ecuaciones lineales Ax=b.

    ! Argumentos de la subrutina
    real(wp), intent(in)    :: a(:,:) ! Matriz del sistema
    real(wp), intent(in)    :: b(:)   ! Término independiente
    real(wp), intent(inout) :: x(:)   ! Aproximación inicial/solución
    real(wp), intent(in)    :: tol    ! Tolerancia para el error
    integer,  intent(inout) :: iter   ! Max iteraciones/iter realizadas
    integer,  intent(out)   :: clave  ! Clave de éxito/error
                                      ! clave  = 0, OK.
                                      ! clave /= 0, max iter. alcanzado
    ! Variables locales
    real(wp) :: diag(size(a,1)) ! Arreglo para guardar la diagonal de A
    real(wp) :: x0(size(x))     ! Arreglo para guardar la iteración anterior
    integer  :: k               ! Contador de iteraciones
    integer  :: i               ! Indice

    ! Procedimiento
    clave = 1
    diag  = [ (a(i,i), i=1,size(a,1)) ]

    do k=1,iter
       x0 = x
       x  = x0 + (b - matmul(a,x0))/diag
       if ( maxval(abs(x-x0)) <= maxval(abs(x))*tol ) then
          clave = 0
          iter  = k
          exit
       end if
    end do

  end subroutine jacobi

  subroutine gauss_seidel(a,b,x,tol,iter,clave)

    ! METODO DE GAUSS SEIDEL para la resolución por iteración del
    ! sistema de ecuaciones lineales Ax = b.

    ! Argumentos de la subrutina
    real(wp), intent(in)    :: a(:,:) ! Matriz del sistema
    real(wp), intent(in)    :: b(:)   ! Término independiente
    real(wp), intent(inout) :: x(:)   ! Aproximación inicial/solución
    real(wp), intent(in)    :: tol    ! Tolerancia para el error
    integer,  intent(inout) :: iter   ! Max iteraciones/iter realizadas
    integer,  intent(out)   :: clave  ! Clave de éxito/error
                                      ! clave  = 0, OK.
                                      ! clave /= 0, max iter. alcanzado
    ! Variables locales
    integer  :: n         ! Dimensión del problema
    integer  :: k         ! Contador de iteraciones
    integer  :: i         ! Indice
    real(wp) :: xi        ! Componente del vector iterado
    real(wp) :: xnorma    ! Norma del vector iterado
    real(wp) :: difnorma  ! Norma de la diferencia entre dos iteraciones

    ! Procedimiento
    clave = 1
    n     = size(a,1)

    do k=1,iter
       xnorma   = 0.0_wp
```

```
      difnorma = 0.0_wp
      do i=1,n
         xi       = (b(i) - dot_product(a(i,1:i-1),x(1:i-1)) &
             & - dot_product(a(i,i+1:n),x(i+1:n)))/a(i,i)
         xnorma   = max(xnorma,abs(xi))
         difnorma = max(difnorma,abs(xi-x(i)))
         x(i)     = xi
      end do
      if (difnorma <= xnorma*tol) then
         iter  = k
         clave = 0
         exit
      end if
   end do

end subroutine gauss_seidel

subroutine jacobi_sparse(ia,ja,a,b,x,tol,iter,info)

   ! METODO DE JACOBI para la resolución por iteración del sistema de ecuaciones
   ! lineales Ax = b donde A es una MATRIZ RALA, la cual es dada en la
   ! representación de coordenadas.
   !
   ! Argumentos:
   !
   ! Input, integer, dimension(:), ia,
   ! Input, integer, dimension(:), ja,
   ! Input, real(wp),dimension(:), a:
   !  arreglos de la representación de coordenadas de la matriz rala.
   !
   ! Input, real(wp), dimension(:), b:
   !  término independiente del sistema de ecuaciones.
   !
   ! Input/Output, real(wp), dimension(:), x:
   !   como dato de entrada es una aproximación inicial a la solución,
   !   como dato de salida es la solución numérica.
   !
   ! Input, real(wp), tol:
   !   tolerancia para el criterio de convergencia.
   !
   ! Input/Output, integer, iter:
   !   como dato de entrada es el número máximo de iteraciones,
   !   como dato de salida es el número de iteraciones realizadas.
   !
   ! Output, integer, info:
   !   Código de error:
   !   info  = 0 => el método convergió,
   !   info /= 0 => el método no convergió después de realizar el
   !                número de máximas iteraciones.

   ! Argumentos
   integer,  intent(in)    :: ia(:)
   integer,  intent(in)    :: ja(:)
   real(wp), intent(in)    :: a(:)
   real(wp), intent(in)    :: b(:)
   real(wp), intent(inout) :: x(:)
   real(wp), intent(in)    :: tol
   integer,  intent(inout) :: iter
   integer,  intent(out)   :: info

   ! Variables locales
   real(wp) :: diag(size(b))
   real(wp) :: x0(size(b))
   integer  :: nelt
   integer  :: k

   info = 1
   nelt = size(ia)

   ! Obtener la diagonal de A
   do k=1,nelt
      if (ia(k) == ja(k) ) then
         diag(ia(k)) = a(k)
      end if
   end do
```

```
    ! Proceder con las iteraciones
    do k=1,iter
       x0 = x
       x = x0 + ( b - matvec_sparse(ia,ja,a,x0) )/diag
       if ( maxval(abs(x-x0)) <= maxval(abs(x))*tol ) then
          info = 0
          iter = k
          exit
       end if
    end do

  end subroutine jacobi_sparse

  function matvec_sparse(ia,ja,a,x) result(y)

    ! matvec_sparse: Computa el vector y = Ax, resultante del producto de
    ! una matriz rala A por un vector x.
    !
    ! Argumentos
    !
    ! Input, integer, dimension(:), ia,
    ! Input, integer, dimension(:), ja,
    ! Input, real(wp),dimension(:), a:
    !  arreglos de la representación de coordenadas de la matriz rala.
    !
    ! Input, real(wp), dimension(:), x,
    !   vector a ser multiplicado por A.
    !
    ! Output, real(wp), dimension(:), y,
    !   vector y = Ax.
    !

    ! Argumentos
    integer,  intent(in) :: ia(:)
    integer,  intent(in) :: ja(:)
    real(wp), intent(in) :: a(:)
    real(wp), intent(in) :: x(:)
    real(wp)             :: y(size(x))

    ! Variables locales
    integer :: k

    ! Procedimiento
    y = 0.0_wp

    do k=1,size(ia)
       y(ia(k)) = y(ia(k))+a(k)*x(ja(k))
    end do

  end function matvec_sparse

end module iterative
```